
Tablas con Stata

L Introduccién

Introduccidn

La presentacién efectiva de los resultados es una parte esencial (y
muchas veces descuidada) del andlisis estadistico.

La presentacién de tablas adecuadas es un elemento importante
dentro de esta comunicacién.

Stata 17 incluye una nueva versién del comando table y el nuevo
conjunto de comandos collect, herramientas que nos permiten
crear tablas con flexibilidad, incluyendo los resultados que
consideremos relevantes, en el orden que querramos, con la
apariencia que querramos.

Tablas con Stata

L Introduccién

Con las nuevas funcionalidades de Stata para tablas, podemos
crear tablas que desplieguen:

o Estadisticos de resumen

@ Resultados de pruebas de hipétesis cldsicas
@ Resultados de regresiones
°

Tests de post-estimacién, medias marginales, efectos
marginales

@ Combinations de los anteriores

@ Resultados retornados por cualquier comando (o
combinaciones de comandos) de Stata.

Tablas con Stata

L Introduccién

Podemos personalizar nuestra tabla
@ Disposicién de la tabla

Formatos numéricos

Apariencia de las etiquetas en filas y columnas
@ Asteriscos y otros textos adicionales

o Fuente, tamaiio y color
°

Sombreado, bordes, margenes, alineamiento, y mds

Tablas con Stata
L Introduccién

Podemos exportar nuestras tablas a
e Word

Excel

ATEX

PDF

Markdown

e HTML

SMCL

Plain text

Podemos incluir la tabla en reportes creados por putdocx,
putexcel, or putpdf.

Tablas con Stata

Descripcién general
e Introduccién a la nueva versién del comando table
@ Introduccién al nuevo conjunto de comandos collect
e Ejemplo: Tablas de resultados de regresion

Tablas con Stata
I—EI comando table

Introduccién al comando table

Tablas con Stata
I—EI comando table

Usaremos algunas variables del conjunto de datos NHANES |1

. use nhanesmod_sp, clear
(Subconjunto de NHANES 2)

. describe
Contains data from nhanesmod_sp.dta
Observations: 10.351 Subconjunto de NHANES 2

Variables: 4 15 Oct 2021 08:43
Variable Storage Display Value

name type format label Variable label
edad byte %9.0g Edad
rural byte %11.0g labrural Rural
grupo_edad byte %8.0g labgrupoedad

Grupo etario

pa_alta byte %8.0g Pres. arterial alta
Sorted by:

Tablas con Stata
I—EI comando table

Las interfaz de usuario para table

table - Table of frequencies, summaries, and command results — *
Main  if/in Weights Statistics Commands Formats  Stars Options

Row variables: (optional)

‘ sex
Column variables: (optional)
[ highbp v
Table variables: (optional)
]

\
Marginal totals
@i
(O None
) Select:

Customize

17| [ & Cancel Submit

Tablas con Stata
I—EI comando table

Elementos basicos

Sintaxis simplificada para table:

table ( variables para filas) ( variables para columnas)

Tablas con Stata
I—EI comando table

L Elementos basicos

Tablas de una via

Por defecto, table construye tablas de frecuencia. Si especificamos
sélo las filas (categorias determinadas por la variable rural), los
resultados se despliegan en una sola columna.

. table (rural) ()

Frequency

Rural
Zona Urbana 6.548
Zona Rural 3.803
Total 10.351

Nota: en general, en Stata, para desplegar los nimeros en formato en
castellano, escribir set dp comma. (revertir con set dp period)

Tablas con Stata
I—EI comando table

L Elementos basicos

Tablas de una via

Si especificamos sélo las columnas, los resultados se despliegan en
una sola fila

. table () (rural)

Rural
Zona Urbana Zona Rural Total

Frequency 6.548 3.803 10.351

Tablas con Stata
I—EI comando table

L Elementos basicos

Tablas de dos vias

Tabulamos variable rural versus grupo_edad

. table (rural) (grupo_edad)

Grupo etario
20-29  30-39 40-49 50-59 60-69 70+ Total

Rural
Zona Urbana 1.631 981 7T 810 1.743 606 6.548
Zona Rural 689 641 495 481 1.117 380 3.803
Total 2.320 1.622 1.272 1.291 2.860 986 10.351

Por defecto, la tabla muestra frecuencias (niimero de observaciones) por celda

y totales por fila y columna

s1ara

I. Cafiette (StataCorp) 15 y 16 de octubre de 2021 13 /77



Tablas con Stata
I—EI comando table

L Elementos basicos

Tablas de dos vias

Podemos suprimir los totales con opcién nototals

. table (rural) (grupo_edad), nototals

Grupo etario
20-29 30-39 40-49 50-59 60-69 70+

Rural
Zona Urbana 1.631 981 77T 810 1.743 606
Zona Rural 689 641 495 481 1.117 380

Tablas con Stata
I—EI comando table

L Elementos basicos

Tablas de dos vias

Podemos desplegar sélo totales por fila o por columna.

. table (rural) (grupo_edad), totals(rural)

Grupo etario
20-29  30-39 40-49 50-59 60-69 70+ Total

Rural
Zona Urbana 1.631 981 77T 810 1.743 606 6.548
Zona Rural 689 641 495 481 1.117 380 3.803

. table (rural) (grupo_edad), totals(grupo_edad)

Grupo etario
20-29  30-39 40-49 50-59 60-69 70+

Rural
Zona Urbana 1.631 981 77T 810 1.743 606
Zona Rural 689 641 495 481 1.117 380
Total 2.320 1.622 1.272 1.291 2.860 986

Tablas con Stata
I—EI comando table

L Elementos basicos

Estadisticos de resumen

La opcién statistic() nos permite tabular, ademds de fecuencias,
otros estadisticos como medias, medianas, proporciones, etc.

table ( variables filas ) ( variables columnas ) ,
statistic( statspec )

Tipos de argumentos para statistic()

nombre genérico | uso

freqstat estadisticos de frecuencia
sumstat estadisticos de resumen
ratiostat estadisticos de proporcién

Tablas con Stata
I—EI comando table

L Elementos basicos

Estadisticos de resumen

statistic(fregstat) produce estadisticos de frequencia.

freqstat Estadistico
frequency | frecuencia (Nimero de observaciones)
sumw suma de los pesos

I. Cafiette (StataCorp) 15 y 16 de octubre de 2021

Tablas con Stata
I—EI comando table

L Elementos basicos

Estadisticos de resumen

statistic(sumstat varlist) despliega estadisticos de resumen para
las variables en varlist.

sumstat Estadistico
mean media
semean error estandar de la media

sebinomial | error estandar de la media, binomial
sepoisson | error estandar de la media, Poisson

variance variancia

sd desviacién estandar

skewness coeficiente de simetria

kurtosis coeficiente de apuntamiento (kurtosis)
cv coefficiente de variacion

Tablas con Stata
I—EI comando table

L Elementos basicos

Estadisticos de resumen

sumstat | Estadistico
count Niumero de valores no faltantes
median | mediana

p# percentil #

ql primer cuartil

q2 segundo cuartil
q3 tercer cuartil
iqr rango intercuartil
min valor minimo
max valor maximo

range rango
primer valor, dltimo, total, y otros

Tablas con Stata
I—EI comando table

L Elementos basicos

Estadisticos de resumen

statistic(ratiostat [varlist]) despliega estadisticos de proporcién.

proportion proporcién
percent porcentaje
rawproportion | proporcién ignorando pesos
rawpercent porcentaje ignorando pesos

Tablas con Stata
I—EI comando table

L Elementos basicos

Estadisticos de resumen

Ejemplo con dos estadisticos en una tabla de una via:

. table () (rural),
> statistic(frequency)
> statistic(percent)

Rural
Zona Urbana Zona Rural Total
Frequency 6.548 3.803 10.351
Percent 63,26 36,74 100,00

Tablas con Stata
I—EI comando table

L Elementos basicos

Estadisticos de resumen

Ejemplo con dos estadisticos en una tabla de dos vias:

. table (rural) (pa_alta),
> statistic(frequency)
> statistic(percent)

Pres. arterial alta
0 1 Total
Rural

Zona Urbana

Frequency 3.798 2.750 6.548

Percent 36,69 26,57 63,26
Zona Rural

Frequency 2.177 1.626 3.803

Percent 21,03 15,71 36,74
Total

Frequency 5.975 4.376 10.351

Percent 57,72 42,28 100,00

Tablas con Stata
I—EI comando table

L Elementos basicos

Estadisticos de resumen

Podemos calcular proporciones a través de filas o columnas con

subopcién across(varnames)
. table (rural) (pa_alta), . table (rural) (pa_alta),
> statistic(frequency) > statistic(frequency)
> statistic(percent, across(pa_alta)) > statistic(percent, across(rural))
> totals(rural) > totals(pa_alta)
Pres. arterial alta Pres. arterial alta
0 1 Total 0 1
Rural Rural
Zona Urbana Zona Urbana
Frequency 3.798 2.750 6.548 Frequency 3.798 2.750
Percent 58,00 42,00 100,00 Percent 63,56 62,84
Zona Rural Zona Rural
Frequency 2.177 1.626 3.803 Frequency 2.177 1.626
Percent 57,24 42,76 100,00 Percent 36,44 37,16
Total
Frequency 5.975 4.376
Percent 100,00 100,00

Tablas con Stata
I—EI comando table

L Elementos basicos

Estadisticos de resumen

Agregamos medias y desviaciones estdndar a la tabla

table (rural) (pa_alta),

> statistic(frequency)
> statistic(percent)
> statistic(mean edad)
> statistic(sd edad)
> nototals
Pres. arterial alta
0 1
Rural
Zona Urbana
Frequency 3.798 2.750
Percent 36,69 26,57
Mean 41,14613 54,69127
Standard deviation 16,77735 15,2568
Zona Rural
Frequency 2.177 1.626
Percent 21,03 15,71
Mean 43,94258 55,44895
Standard deviation 16,61634 14,29346 ,

Tablas con Stata
I—EI comando table

L Elementos basicos

Cambio de formato del estadistico

Podemos cambiar los formatos de los estadisticos con la sintaxis:

table ( variables filas ) ( variables columnas ) ,
nformat() sformat()

@ nformat () - Formato numérico; escribir help format para
ver las especificaciones disponibles.

o sformat () - Formato alfanumérico, donde %s se sustituye por
el valor del estadistico dentro de un texto entre comillas. (por
ejemplo para un precio, podemos escribir sformat ("$%s") )

Tablas con Stata
I—EI comando table

L Elementos basicos

. collect clear
table (rural) (pa_alta),

> statistic(frequency)

> statistic(percent)

> statistic(mean edad)

> statistic(sd edad)

> nototals

> nformat (%9.0fc frequency)

> sformat ("%s%%" percent)

> nformat (%6.2f mean sd)

> sformat (" (%s)" sd)

Pres. arterial alta
0 1
Rural
Zona Urbana
Frequency 3.798 2.750
Percent 36,69% 26,57%
Mean 41,15 54,69
Standard deviation (16,78) (15,26)
Zona Rural

Frequency 2.177 1.626
Percent 21,03% 15,71%
Mean 43,94 55,45
Standard deviation (16,62) (14,29)

Tablas con Stata
I—EI comando table

I—E><tensiones de comando table

Resultados de otros comandos

También podemos usar table para tabular resultados de otros

comandos.
La sintaxis es:

table ( rowspec ) ( colspec),
command ( cmdspec )

Tablas con Stata
I—EI comando table

I—E><tensiones de comando table

Resultados de otros comandos

Por ejemplo:

table (...) (...),
command (correlate edad rural pa_alta)

Tablas con Stata
I—EI comando table

I—E><tensiones de comando table

correlate guarda las correlaciones en la matriz r(C).

. correlate edad rural pa_alta
(obs=10.351)

edad rural pa_alta
edad 1,0000
rural 0,0568 1,0000
pa_alta 0,3676 0,0074 1,0000
. return list
scalars:
r(N) = 10351
r(rho) = ,0567812234319929
matrices:

r(¢) : 3x3
. matrix list r(C)

symmetric r(C)[3,3]

edad rural pa_alta
edad 1
rural .05678122 1
pa_alta .36755099 .00739899 1

Tablas con Stata
I—EI comando table

I—E><tensiones de comando table

Construimos una tabla con la matriz de correlacion

. table (rowname) (colname),

> command (r(C) : correlate edad rural pa_alta) nformat(%5.4f)
Edad Rural Pres. arterial alta

Edad 1,0000 0,0568 0,3676

Rural 0,0568 1,0000 0,0074

Pres. arterial alta 0,3676 0,0074 1,0000

Tablas con Stata

I—EI conjunto de comandos comandos collect

Introduccidén al uso de collect

Tablas con Stata

I—EI conjunto de comandos comandos collect

Elementos basicos: Construir tablas con collect

@ Recolectar resultados de otros comandos: comando —collect
get, prefijo collect:, o table

@ Explorar la collecion—collect dims, collect levelsof, y
collect label list

© Definir las filas y las and columnas de la tabla—collect
layout o table

@ Personalizar la tabla, especificar formatos, etiquetas, fuente,
sombreado, y mdas—collect label, collect style, collect
stars, ...

@ Exportar la tabla a Word, Excel, IATEX, PDF, Markdown,
HTML, SMCL, o simple texto—collect export

@ Grabar el estilo, etiquetas, y la coleccién utilizada para seguir
modificando mds tarde o para utilizar el estilo con otra
coleccién—collect label save, collect style save, collect

save sSTata

Tablas con Stata

I—EI conjunto de comandos comandos collect

Interfaz para construir tablas (Tables Builder)

Tables Buider

Collection: | Table

Dimensions

Levels

High biood pressure (highbp)
Soc sex

Mean (mean)
tandard deviation (sd)

Rows.

sexcresutt v

Frequency (frequency)
Statstc option variable (var) |Percent (percent)
Across (across)

Coveriate names and column n.

Command results index (e,

Statstc/command option inde.

[Command option index (com.

Teble celtype (celltype)

Teble border block (border_blo.

Label and style dilogs

Remap tags

Columnz
Highop
Tables
+
it dmension labes
Editleve labels o
THigh blood pressure
Compos rowheaders o :
Compote cobumn headers =
Compose able header: s
Frequency
Show/hide header content Peichit 2522%
Cell sppesrance syles Mean 4230
Standard deviation|  (16.97)  (15.88)
Intercept postion e raly
‘Show/hide coefficient styles Frequency 3364 2072
Percent 3250% 20.02%
Automstic dimension evels o e ap
Recode dimenion levels Standard deviation| (16.60) _(13.26)

Export

ette (StataCorp)

y 16 de octubre de 2021

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 1: Recolectar resultados

El primer paso es poner los resultados de interés en una coleccién
o Prefijo collect—Recolecta los resultados de command
collect: command

@ collect get—Recolecta resultados especificos del comando
previo

command
collect get results
@ table— Pone los resultados en una coleccién
automaticamente.

Comenzamos usando la coleccién que creamos con table.

Tablas con Stata

I—EI conjunto de comandos comandos collect

. table (rural) (pa_alta),

> statistic(frequency)

> statistic(percent)

> statistic(mean edad)

> statistic(sd edad)

> nototals

> nformat (%9.0fc frequency)

> sformat ("%s%%" percent)

> nformat (%6.2f mean sd)

> sformat (" (%s)" sd)

Pres. arterial alta
0 1
Rural
Zona Urbana
Frequency 3.798 2.750
Percent 36,69% 26,57%
Mean 41,15 54,69
Standard deviation (16,78) (15,26)
Zona Rural

Frequency 2.177 1.626
Percent 21,03% 15,71%
Mean 43,94 55,45
Standard deviation (16,62) (14,29)

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 2: Explorar la coleccién

Los valores en una coleccién estdn organizados en lo que llamamos
tags.

Una tag consiste en una dimensién (dimension) y un nivel (level)
dentro de la dimensidén. Una tag se especifica indicando dimensién
y nivel (dimension[level]).

Valor | Tag 1 Tag 2 Tag 3

3.789 | rural[1] | pa_altal0] | result[frequencyl]
1,625 | rural[1] | pa_alta[1] | result[frequency]
21,03 | rural[1] | pa_alta[0] | result[percent]

Las dimensiones involucradas en la tabla son rural, pa_alta, y

result, por lo que son parte de la coleccidn, junto con otras

dimensiones. ‘
sT1ara

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 2: Explorar la coleccién

collect dims lista todas las dimensiones en una coleccion.

. collect dims

Collection dimensions
Collection: Table

Dimension No. levels

Layout, style, header, label
across
cmdset

colname
command
pa_alta
result
rural
statcmd
var

NN NE = =N

Style only
border_block
cell_type

o

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 2: Explorar la coleccién

collect levelsof lista los niveles de la dimensién especificada.

. collect levelsof pa_alta

Collection: Table
Dimension: pa_alta
Levels: 0 1

collect label list lista los niveles y etiquetas asociadas a una
dimension.

. collect label list pa_alta, all

Collection: Table

Dimension: pa_alta
Label: Pres. arterial alta
Level labels:
.m Total
0
1

Por defecto, la dimensién asociada a una variable y sus nivelas
heredan las etiquetas de la variable, pero las podemos cambiar.

I. Cafiette (StataCorp) 15 y 16 de octubre de 2021 38 /77



Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 3: Especificar filas y columnas

Hemos especificado filas y columnas con table. Desplegamos dicha
disposicién con collect layout .
Como estamos represen-

- collect layout tando tres dimensiones

Collection: Table

Rows: rural#result (pa_alta, rural Yy result en
Columns: pa_alta .
rople 1) Gi e filas y columnas, (en este
caso) las filas contienen
Pres. arterial alta . .,
0 1 la combinacién de dos
Rural dimensiones.
Zona Urbana
Frequency 3.798 2.750
Percent 36,697 26,57%
Mean 41,15 54,69
Standard deviation (16,78) (15,26)
Zona Rural
Frequency 2.177 1.626
Percent 21,039 15,71%
Mean 43,94 55,45 .
Standard deviation (16,62) (14,29) sSTata

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 3: Especificar filas y columnas

collect layout también nos permite establecer la disposicidn

. collect layout (result) (rural#pa_alta)
Collection: Table
Rows: result
Columns: rural#pa_alta
Table 1: 4 x 4

Rural
Zona Urbana Zona Rural
Pres. arterial alta Pres. arterial alta
0 1 (] 1
Frequency 3.798 2.750 2.177 1.626
Percent 36,69% 26,57% 21,03% 15,71%
Mean 41,15 54,69 43,94 55,45
Standard deviation (16,78) (15,26) (16,62) (14,29)

s1ara

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 3: Especificar filas y columnas

Podemos especificar los niveles que queremos tabular para las
distintas dimensiones

. collect layout (result[mean sd]) (rural#pa_alta)

Collection: Table
Rows: result[mean sd]
Columns: rural#pa_alta
Table 1: 2 x 4
Rural
Zona Urbana Zona Rural
Pres. arterial alta Pres. arterial alta
0 1 0 1
Mean 41,15 54,69 43,94 55,45
Standard deviation (16,78) (15,26) (16,62) (14,29)

I. Cafiette (StataCorp)

Tablas con Stata

I—EI conjunto de comandos comandos collect

Volvemos a la tabla original:
. collect clear
table (rural) (pa_alta),

> statistic(frequency)

> statistic(percent)

> statistic(mean edad)

> statistic(sd edad)

> nototals

> nformat (%9.0fc frequency)

> sformat ("%s%%" percent)

> nformat (%6.2f mean sd)

> sformat (" (%s)" sd)

Pres. arterial alta
0 1
Rural
Zona Urbana
Frequency 3.798 2.750
Percent 36,69% 26,57%
Mean 41,15 54,69
Standard deviation (16,78) (15,26)
Zona Rural

Frequency 2.177 1.626
Percent 21,03% 15,71%
Mean 43,94 55,45
Standard deviation (16,62) (14,29) sTata

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 4: Personalizar la tabla

Con collect label dim especificamos la etiqueta para una
dimensién.

Con collect label levels especificamos etiquetas para los niveles
de una dimensién. Especificamos etiquetas de dimensién y niveles

para pa_alta

. collect label dim pa_alta "Hipertensién", modify
. collect label levels pa_alta O "No" 1 "Si"
. collect label list pa_alta, all
Collection: Table
Dimension: pa_alta
Label: Hipertensién
Level labels:
.m Total
0 No
1 si

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 4: Personalizar la tabla

Especificamos etiquetas de dimensién y niveles para rural

. collect label dim rural "Zona", modify
. collect label levels rural O "Urbana" 1 "Rural", modify
. collect label list rural, all

Collection:
Dimension:
Label:

Level labels:

.m
0
1

Table
rural
Zona

Total
Urbana
Rural

s1ara

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 4: Personalizar la tabla

collect preview nos muestra la tabla con las nuevas etiquetas.

. collect preview

Hipertensién
No Si
Zona
Urbana
Frequency 3.798 2.750
Percent 36,69% 26,57%
Mean 41,15 54,69
Standard deviation (16,78) (15,26)
Rural
Frequency 2.177 1.626
Percent 21,083% 15,71Y%
Mean 43,94 55,45
Standard deviation (16,62) (14,29)

s1ara

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 4: Personalizar la tabla

También podemos modificar las etiquetas de valores para los
resultados; las desplegamos con collect label list

. collect label list result

Collection:
Dimension:
Label:

Level labels:

frequency
mean
percent
sd

Table
result
Result

Frequency

Mean

Percent

Standard deviation

s1ara

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 4: Personalizar la tabla

... y las modificamos con collect label levels

collect label levels result

vV V.V V.

. collect preview

frequency "Frecuencia"

Zona
Urbana
Frecuencia
Porcentaje
Media (edad)
Desviacién estandar (edad)
Rural
Frecuencia
Porcentaje
Media (edad)
Desviacién estandar (edad)

mean "Media (edad)"
percent "Porcentaje"
sd "Desviacién estandar (edad)",
replace
Hipertensién
No Si

3.798 2.750

36,69% 26,57%

41,15 54,69

(16,78) (15,26)

2.177 1.626
21,037 15,71%
43,94 55,45

(16,62) (14,29)

s1ara

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 4: Personalizar la tabla

Los subcomandos collect style nos permiten hacer muchos tipos
de modificaciones en la apariencia de la tabla. Aqui removemos el
borde a la derecha de los encabezados de fila.

. collect style cell border_block, border(right, pattern(nil))

. collect preview

Hipertensién
No Si
Zona
Urbana
Frecuencia 3.798 2.750
Porcentaje 36,69% 26,57%
Media (edad) 41,15 54,69
Desviacién estandar (edad) (16,78) (15,26)
Rural
Frecuencia 2.177 1.626
Porcentaje 21,03% 15,71%
Media (edad) 43,94 55,45
Desviacién estandar (edad) (16,62) (14,29) STaTa

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 5: Exportar la tabla

collect export exporta la tabla personalizada al formato que
especificamos. chosen format.

Cuando exportamos a Word, collect style putdocx nos permite
hacer cambios especificos para este formato.

. collect style putdocx, layout(autofitcontents)
. collect export mitabla.docx, replace
(collection Table exported to file mitabla.docx)

s1ara

I. Cafiette (StataCorp) 15 y 16 de octubre de 2021 49 /77



Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 5: Exportar la tabla

Table T... m - 0O

View | Help | Table Design  Layout @ Telime £ Share

AaBbceDc | AaBbC¢ AaBbCc w || (i)

Faste WA A S TNormal | Heading1 Heading2 |<| | Edting
Clipboard % Font o Paragraph [ Styles ™ ~
Hipertension

No Si
Zona
Urbana
Frecuencia 3.798 2.750
Porcentaje 36.69% 26.57%
Media (edad) 41,1461295 54.6912727
Desviacion estandar (edad) (16.78) (15.26)
Rural
Frecuencia 2177 1.626
Porcentaje 21.03% 15.71%
Media (edad) 43,9425815  55,4489545
Desviacién estandar (edad) (16.62) (14.29)
Pageiof1 36words [ {9 Accessibility: Unavailable E B ———8F—+ 120% sTaTa

Tablas con Stata

I—EI conjunto de comandos comandos collect

Paso 5: Exportar la tabla

Exportamos la tabla a IATEX; la linea siguiente genera un archivo
.tex que se puede compilar directamente para generar un archivo
con la tabla.

collect export mitablaO.tex

La opcién tableonly nos permite exportar solamente el cédigo que
genera la tabla,
collect export mitabla.tex, tableonly

asi podemos incluir la tabla en otro archivo en el que estamos
trabajando, por ejemplo:

s1ara

Tablas con Stata

I—EI conjunto de comandos comandos collect

\documentclass{article}

% los siguientes paquetes se requieren para tablas
\usepackage{multirow}

\usepackage{amsmath}

\usepackage{ulem}

\usepackage [table]{xcolor}

% El1 siguiente paquete es para que LaTeX entienda Unicode
% asi podemos usar el teclado en castellano.
\usepackage [utf8] {inputenc}

\begin{document}

\large{\bf Podemos poner algin titulo}
Aqui podemos incorporar algin texto:
(... texto omitido)

\vspace{2ex}

\begin{center}
\input{mitabla}
\end{center}

Aqui podemos seguir con nuestro documento.
\end{document}

Tablas con Stata

I—EI conjunto de comandos comandos collect

eoe midocumento.dvi (page 1 of 1)
1 Q @@ A O8I+ » 0 00
Previous/Next  Page  Back/Forward Zoom Tool Mode Add Note

Podemos poner algin titulo

Aqui podemos incorporar algin texto: por ejemplo, aprovecho a
sugerir que la primera vez que exportamos una tabla a ITgXpodemos
hacerlo sin la opcién tableonly, asi vemos cuéles son los paque-
tes requeridos para desplegar debidamente una tabla creada con

collect.
Hypertensién
No Si
Zona
Urbana
Frecuencia 3.798 2.750
Porcentaje 36,69% 26,57%
Media (edad) 4115 54,69
Desviacién estandar (edad) (16,78) (15,26)
Rural
Frecuencia 2.177 1.626
Porcentaje 21,03% 15,71%
Media (edad) 43,94 5545

Desviacién esténdar (edad) (16,62) (14,29)

Aqui podemos seguir con nuestro documento.

Tablas con Stata

I—Ejemplos con resultados de estimacién

Ejemplo: Tabla de resultados de regresion

Queremos crear una tabla que incluya los resultados de

una

regresion logistica como razén de momios (odds ratios) y sus

errores estandar.

. logistic pa_alta c.edad i.rural

Logistic regression Number of obs = 10.351
LR chi2(2) = 1464,14
Prob > chi2 = 0,0000
Log likelihood = -6318,6948 Pseudo R2 = 0,1038
pa_alta | Odds ratio Std. err. z P>|z| [95% conf. intervall
edad 1,048494 ,0013846 35,86 0,000 1,045784 1,051212

rural
Zona Rural ,9426129 ,0418033 -1,33 0,183 ,8641398 1,028212
_cons ,0744018 ,0052447 -36,86 0,000 ,0648009 ,0854253

Note: _cons estimates baseline odds.

I. Cafiette (StataCorp) 15 y 16 de octubre de 2021

Tablas con Stata

I—Ejemplos con resultados de estimacién

También queremos incluir los indices AIC y BIC.

. estat ic

Akaike’s information criterion and Bayesian information criterion

Model N 11(null) 11l(model) df AIC BIC

10.351 -7050,765 -6318,695 3 12643,39 12665,12

Note: BIC uses N = number of observations. See [R] BIC note.
. return list

matrices:
r(s) : 1x6

. mat list r(S)

r(s)[1,6]
N 110 11 af AIC BIC
10351 -7050.7655 -6318.6948 3 12643.39 12665.124
. display r(s)[1,"AIC"]
12643,39

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

Los comandos de Stata graban resultados en r() (como vimos para
correlate y estatic) y/o en e() (comandos de estimacién). La
lista y la descripcidén de estos resultados estdn en la ayuda del
comando correspondiente. (“Stored results”). Luego de correr un
comando, los vemos con

return list
return list

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

. logistic pa_alta c.edad i.rural
(salida omitida)

. ereturn list

scalars:
e(rank) = 3
e(N) = 10351
...
e(11l) = -6318,6948199982
C...)
macros:
e(cmdline) : "logistic pa_alta c.edad i.rural"
e(cmd) : "logistic"
C...)
matrices:
e() : 1x4
e(V) : 4x 4
e(Cns) : 1 x5
...
functions:

e(sample)
S1ata

Tablas con Stata

I—Ejemplos con resultados de estimacién

Por defecto, el prefijo collect: recolecta todos los resultados en
e() o r(). La excepcidn son las matrices e(b) y e(V) (aunque las
podemos recolectar si queremos). Los resultados de la tabla de
estimacién se recolectan bajo los nombres _r_b, _r_se, r_z, _r_p,
etc. (escribir . help collect get)

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

Creamos la coleccién Mismodelos y alli ponemos nuestros

resultados de regresién y le asignamos la tag modelo[(1)]. Aqui

estamos creando las dos cosas, una dimensién (modelo) y un nivel
(1). (aqui estamos recolectando sélo coeficientes y errores

estandar)

. collect clear

. collect create Mismodelos
(current collection is Mismodelos)

. collect _r_b _r_se, tag(modelo[(1)])

: logistic pa_alta c.edad i.rural

Logistic regression Number of obs = 10.351
LR chi2(2) = 1464,14
Prob > chi2 = 0,0000
Log likelihood = -6318,6948 Pseudo R2 = 0,1038
pa_alta | Odds ratio Std. err. z P>|z| [95% conf. intervall
edad 1,048494 ,0013846 35,86 0,000 1,045784 1,051212
rural
Zona Rural ,9426129 ,0418033 -1,33 0,183 ,8641398 1,028212
_cons ,0744018 ,0052447 -36,86 0,000 ,0648009 ,0854253
I-, Cafiette (StataCorp) 15 y 16 de octubre de 2021

Tablas con Stata

I—Ejemplos con resultados de estimacién

Agregamos los resultados de estat ic a la coleccién, también
asignandole la dimensién modelo y el nivel (1).

. collect AIC=r(S)[1,"AIC"] BIC=r(S)[1,"BIC"], tag(modelo[(1)]) : estat ic

Akaike”s information criterion and Bayesian information criterion

Model N 11(null) 11l(model) df AIC BIC

10.351 -7050,765 -6318,695 3 12643,39 12665,12

Note: BIC uses N = number of observations. See [R] BIC note.

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

En total, recolectamos resultados de dos modelos
. collect clear
. collect create Mismodelos

. collect _r_b _r_se, tag(modelo[(1)]): logistic pa_alta c.edad i.rural
. collect AIC=r(S)[1,5] BIC=r(S)[1,6], tag(modelo[(1)]): estat ic

. collect _r_b _r_se, tag(modelo[(2)]): logistic pa_alta c.edad##i.rural
. collect AIC=r(S)[1,5] BIC=r(S)[1,6], tag(modelo[(2)]): estat ic

A continuacién especificamos la disposicion de nuestra tabla:
ponemos variables independientes (colnames) y estadisticos
(result) en las filas y la dimensién modelo en las columnas.

. collect layout (colname#result) (modelo)

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

. collect layout (colname#result) (modelo)

Collection: Mismodelos
Rows: colname#result
Columns: modelo
Table 1: 18 x 2

(6] )

Edad

Coefficient 1,048494 1,049813

Std. error ,0013846 ,0017259
Zona Urbana

Coefficient 1 1

Std. error 0 0
Zona Rural

Coefficient ,9426129 1,129659

Std. error ,0418033 ,1650808
Zona Urbana # Edad

Coefficient 1

Std. error 0
Zona Rural # Edad

Coefficient ,9964202

Std. error ,0027514
Intercept

Coefficient ,0744018 ,069929

Std. error ,0052447 ,0059844 sTata

Tablas con Stata

I—Ejemplos con resultados de estimacién

Omitimos los niveles de base en la tabla.

. collect style showbase off
. collect preview

) (2)

Edad

Coefficient 1,048494 1,049813

Std. error ,0013846 ,0017259
Zona Rural

Coefficient ,9426129 1,129659

Std. error ,0418033 ,1650808
Zona Rural # Edad

Coefficient ,9964202

Std. error ,0027514
Intercept

Coefficient ,0744018 ,069929

Std. error ,0052447 ,0059844

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

Especificamos el formato numérico de los resultados. (Hemos las
opciones nformat() y sformat() directamente con el comando
table; cuando constrimos la tabla con collect, esas son opciones
para collect style cell.

. collect style cell, nformat(%5,2f)
. collect style cell result[_r_sel, sformat("(%s)")

. collect preview

(&) (2)

Edad

Coefficient 1,05 1,05

Std. error (0,00) (0,00)
Zona Rural

Coefficient 0,94 1,13

Std. error (0,04) (0,17)
Zona Rural # Edad

Coefficient 1,00

Std. error (0,00)
Intercept

Coefficient 0,07 0,07

Std. error (0,01) (0,01)

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

Removemos el borde a la derecha de las etiquetas de fila.

. collect style cell border_block, border(right, pattern(nil))
. collect preview

[¢) (2)

Edad

Coefficient 1,05 1,05

Std. error (0,00) (0,00)
Zona Rural

Coefficient 0,94 1,13

Std. error (0,04) (0,17)
Zona Rural # Edad

Coefficient 1,00

Std. error (0,00)
Intercept

Coefficient 0,07 0,07

Std. error (0,01) (0,01)

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

Centramos los encabezados de columna (column-header) y los
resultados (item)

. collect style cell cell_typelitem column-header], halign(center)

. collect preview

) (2)

Edad

Coefficient 1,05 1,05

Std. error (0,00) (0,00)
Zona Rural

Coefficient 0,94 1,13

Std. error (0,04) (0,17)
Zona Rural # Edad

Coefficient 1,00

Std. error (0,00)
Intercept

Coefficient 0,07 0,07

Std. error (0,01) (0,01)

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

Ocultamos las etiquetas para los estadisticos.

. collect style header result, level(hide)

. collect preview

(&) (2)

Edad 1,05 1,05
(0,00) (0,00)

Zona Rural 0,94 1,13
(0,04) (0,17)

Zona Rural # Edad 1,00
(0,00)

Intercept 0,07 0,07

(0,01) (0,01)

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

Agregamos un espacio entre columnas.

. collect style column, extraspace(1l)
. collect preview

(D] (2)

Edad 1,05 1,05
(0,00)  (0,00)

Zona Rural 0,94 1,13
(0,04)  (0,17)

Zona Rural # Edad 1,00
(0,00)

Intercept 0,07 0,07

(0,01)  (0,01)

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

Agregamos un espacio entre filas y usamos una x para indicar
interacciones.

. collect style row stack, spacer delimiter(" x ")
. collect preview

(6D (2)

Edad 1,05 1,05
(0,00)  (0,00)

Zona Rural 0,94 1,13
(0,04) (0,17)

Zona Rural x Edad 1,00
(0,00)
Intercept 0,07 0,07

(0,01)  (0,01)

s1ara

I. Cafiette (StataCorp) 15 y 16 de octubre de 2021 69 / 77



Tablas con Stata

I—Ejemplos con resultados de estimacién

Agregamos los AIC y BIC al final de cada columna.

. collect layout (colname#result result[AIC BIC]) (modelo)
Collection: Mismodelos
Rows: colname#result result[AIC BIC]
Columns: modelo
Table 1: 13 x 2

(6] (2)

Edad 1,05 1,05
(0,00) (0,00)

Zona Rural 0,94 1,13
(0,04) (0,17)

Zona Rural x Edad 1,00
(0,00)

Intercept 0,07 0,07
(0,01) (0,01)

12643,39 12643,71
12665,12 12672,69

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

Agregamos etiquetas para AlC y BIC.

. collect style header result[AIC BIC], level(label)

. collect preview

(D] (2)
Edad 1,05 1,05
(0,00) (0,00)
Zona Rural 0,94 1,13
(0,04) (0,17)
Zona Rural x Edad 1,00
(0,00)
Intercept 0,07 0,07
(0,01) (0,01)
AIC 12643,39 12643,71
BIC 12665,12 12672,69

I. Cafiette (StataCorp)

Tablas con Stata

I—Ejemplos con resultados de estimacién

Especificamos otro formato para AlC y BIC.

. collect style cell result[AIC BIC], nformat(%8,0f)
. collect preview

(6D (2)

Edad 1,05 1,05
(0,00)  (0,00)

Zona Rural 0,94 1,13
(0,04) (0,17)

Zona Rural x Edad 1,00
(0,00)
Intercept 0,07 0,07

(0,01)  (0,01)
AIC 12643 12644

BIC 12665 12673

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

Grabar los estilos y etiquetas (Les debia el Paso 6)

Después de personalizar una tabla, podemos grabar los estilos y
etiquetas para usarlos luego en otras tablas.

. collect style save style_regl, replace
(style from Mismodelos saved to file style_regl.stjson)

. collect label save label_regl, replace
(labels from Mismodelos saved to file label_regl.stjson)

s1ara

Tablas con Stata

I—Ejemplos con resultados de estimacién

Ahora recolectamos resultados de modelos logisticos aplicados a
otro conjunto de datos. Especificamos las filas y columnas, y
aplicamos los estilos que hemos grabado.

. collect

. use lbw.

. collect
. collect

. collect
. collect
. collect

. collect
. collect

clear
sp, clear

rb _r_se, tag(modelo[(1)]): logistic bajo-peso c.edad i.fuma
AIC=r(S)[1,5] BIC=r(S)[1,6], tag(modelo[(1)]): estat ic

rb _r_se, tag(modelo[(2)]): logistic bajo.peso c.edad##i.fuma
AIC=r(S)[1,5] BIC=r(S)[1,6], tag(modelo[(2)]1): estat ic
layout (colname#result result[AIC BIC]) (modelo)

style use style_regl, override
label use label_regl

I. Cafiette (StataCorp) 15 y 16 de octubre de 2021 74 )77



Tablas con Stata

I—Ejemplos con resultados de estimacién

. collect preview

(6] (2)

Edad 0,95 0,92
(0,03)  (0,04)

Fuma 2,00 0,38
(0,64) (0,58)

Fuma x Edad 1,08
(0,07)

Intercept 1,06 2,24
(0,80)  (2,29)

AIC 233 234
BIC 243 247

I. Cafiette (StataCorp)

15 y 16 de octubre de 2021

Tablas con Stata

L Comentarios finales

@ El comando table nos permite crear facilmente nuestras
tablas con estadisticos de resumen, y resultados de otros
comandos de Stata.

@ El conjunto de comandos collect también nos permite
construir tablas simples o complejas, personalizarlas, y
exportarlas a varios formatos para incorporarlas directamente
en nuestros reportes.

@ Podemos guardar estilos y etiquetas en disco, para luego
usarlos en nuevas tablas en el futuro.

s1ara

Tablas con Stata
I—Comentarios finales

Recursos adicionales:

@ Por preguntas adicionales, pueden escribirnos a
tech-support@stata.com
@ Documentacién
o table
https://www.stata.com/manuals/rtableintro.pdf
o collect
https://www.stata.com/manuals/tables.pdf
o Videos de Stata (sobre éste y otros temas)
https://www.stata.com/links/video-tutorials/

@ Otros recursos oficiales de Stata:
https://www.stata.com/support/

e Foro de Stata (moderado por usuarios)
https://www.statalist.org/

s1ara

